


Jeux et calculatrices : les affichages combinés


L'agrément d'un jeu programmé tient en partie à la façon dont il est présenté. C'est dire qu'il faut soigner les messages affichés. Voici de nouvelles techniques que vous pourrez adapter à vos propres jeux (exemple sur TI-57).

■ Le premier jeu que nous vous proposons est une illustration typique d'une astuce d'affichage. Je l'ai baptisé *Les Huns*, mais il n'a heureusement rien à voir avec les hor-

des sauvages qui déferlèrent jadis sur l'Europe, en rasant tout sur leur passage. C'est en fait l'adaptation d'un jeu très classique dans lequel on doit deviner, en un minimum d'essais, un nombre compris entre 1 et 100 tiré par la machine, en s'aidant des seules indications — plus petit ou plus grand — qu'elle fournit, ces indications apparaissent sous forme de l'affichage de +1 ou de -1. Dans *Les Huns*, cette indication sera donnée par l'affichage d'une série plus ou moins longue de 1 qui représenteront symboliquement la « distance » séparant le nombre proposé du nombre à découvrir. A remarquer que le joueur ne sait pas lequel des deux nombres est le plus grand, ce qui ajoute un peu de sel au jeu.

Le nombre d'essais effectués sera





affiché en permanence grâce à une technique connue d'affichage simultané : on utilisera les deux chiffres de l'exposant. La véritable difficulté consiste ici à obtenir la conversion automatique en notation scientifique-


que. En notation scientifique en effet, la mantisse ne peut comporter qu'un seul chiffre avant la virgule. Ainsi un résultat comme 111 05, où 111 représente symboliquement l'écart entre le nombre proposé et celui qu'il faut découvrir (5 étant le nombre d'essais), sera traduit par la machine en 1.11 07 : le nombre d'essais n'est plus exact.

Comme le nombre de chiffres pouvant apparaître derrière la virgule est ici variable, le seul remède efficace consiste à mesurer la longueur du nombre affiché, pour en tenir compte et rectifier l'affichage du nombre d'essais en conséquence.

La technique permettant de mesurer la longueur du nombre (en pratique ici, la longueur diminuée de 1) peut être utilement exploitée pour d'autres jeux : on prendra simplement la partie entière du logarithme



C'EST TOI QUI LE DIS!



Les Huns

Programme pour TI-57

Auteur Jacques Deconchat

Copyright l'Ordinateur de poche et l'auteur

00	19		2nd	Ct
01	33	1		RCL 1
02	-18		2nd	Inv Log
03	-49		2nd	Inv Int
04	32	1		STO 1
05	55			X
06	02			2
07	-18		2nd	Inv Log
08	75			+
09	01			1
10	85			=
11	49		2nd	Int
12	38	7	2nd	Exc 7
13	32	3		STO 3
14	86	3	2nd	Lbl 3
15	81			R/S
16	-42			Inv EE
17	66		2nd	x = t
18	51	9		GTO 9
19	65			-
20	33	7		RCL 7
21	85			=
22	40		2nd	x
23	24			√
24	49		2nd	Int
25	32	0		STO 0
26	01			1
27	34	3		SUM 3
28	86	1	2nd	Lbl 1
29	33	0		RCL 0
30	-18		2nd	Inv Log
31	75			+
32	56		2nd	dsz
33	51	1		GTO 1
34	01			1
35	85			=
36	45			÷
37	18		2nd	Log
38	49		2nd	Int
39	-18		2nd	Inv Log
40	55			x
41	33	3		RCL 3
42	-18		2nd	Inv Log
43	85			=
44	42			EE
45	51	3		GTO 3

Jeux et calculatrice :

les affichages combinés

Utilisation du programme
Initialiser avec 0xxxx STO 1
Début de partie : RST puis R/S
On joue en tapant n R/S

décimal. En effet $\text{Log}(10) 1 = 0$, car $10^0 = 1$; $\text{Log}(10) 10 = 1$, car $10^1 = 10$; $\text{Log}(10) 100 = 2$, car $10^2 = 100$, etc.

Ainsi tout nombre compris entre 1 et 10 aura son logarithme décimal compris entre 0 et 1, tout nombre entre 10 et 100 aura son logarithme entre 1 et 2, etc. L'affichage recherché sera donc programmé par la formule $n : 10^{\text{int}(\text{Log}_{10}n)} \times 10M = EE$ (M étant le nombre d'essais). L'instruction EE est utilisée pour forcer le passage en notation exponentielle qui n'est pas automatique pour les nombres inférieurs à 10^8 .

On trouvera ci-dessus le mode d'utilisation du programme et ci-dessous un exemple de partie.

Exemple de partie

Faire 0.52134 STO 1
RST
R/S affichage : 0
50 R/S " : 1.1111 01
60 R/S " : 1.1111 02
30 R/S " : 1.1 03
35 R/S " : 1.1 04
27 R/S " : 1.11 05
33 R/S " : 33 (clignotant)

On a gagné, le nombre à découvrir était bien 33. La séquence CLR RCL 3 rappelle à l'affichage le nombre d'essais infructueux (ici 5).

Quand il faut obtenir zéro

Le second jeu, intitulé *Remise à zéro*, fait aussi appel à une technique d'affichage élaborée. Il utilise par ailleurs un contrôle du coup joué qui n'est pas complet, mais qui rend plus sûr le déroulement des parties.

Au départ, la machine affiche un nombre en notation scientifique tel que 9.1234567 08 ou 9.1234567 05. Le premier chiffre (9) indique le nombre d'essais auxquels le joueur a droit. La partie fractionnaire de la

mantisse (1234567) est composée des chiffres qu'il va falloir éliminer, on verra comment dans un moment. L'exposant enfin est un nombre tiré au hasard et compris entre 2 et 12.

Remise à zéro

Programme pour TI-57
Auteur Jacques Deconchat
Copyright l'Ordinateur de poche et l'auteur.

00	33	0	RCL 0
01	75		+
02	33	6	RCL 6
03	85		=
04	55		x
05	33	7	RCL 7
06	-18	2nd	Inv Log
07	42		EE
08	85		=
09	81		R/S
10	-34	7	Inv SUM 7
10	55		x
12	84		+/-
13	-18	2nd	Inv Log
14	85		=
15	34	5	SUM 5
16	71		RST
17	86	1	2nd Lbl 1
18	33	4	RCL 4
19	-56	2nd	Inv dsz
20	51	9	GTO 9
21	61	6	SBR 6
22	32	7	STO 7
23	61	6	SBR 6
24	34	7	SUM 7
25	15		CLR
26	32	5	STO 5
27	71		RST
28	86	6	2nd Lbl 6
29	33	1	RCL 1
30	-18	2nd	Inv Log
31	-49	2nd	Inv Int
32	32	1	STO 1
33	55		x
34	06		6
35	75		+
36	01		1
37	85		=
38	49	2nd	Int
39	-61		Inv SBR
40	86	0	2nd Lbl 0
41	15		CLR
42	-66	2nd	Inv x = t
43	51	1	GTO 1
44	33	5	RCL 5
45	-34	6	Inv SUM 6
46	33	6	RCL 6
47	-66	2nd	Inv x = t
48	51	1	GTO 1
49	33	3	RCL 3

Utilisation du programme

Initialiser avec 0.xxxx STO 1
3.01837 STO 4
3.7183037 STO 3
10 STO 0
0.1234567 STO 6

Les deux dernières valeurs doivent être réintroduites avant chaque partie.

On commence avec SBR 1 et l'on indique les chiffres à enlever en tapant n_i R/S. La remise à zéro de ces chiffres s'obtient avec SBR 0. Si le coup n'est pas possible, on passe un tour avec SBR 1.

Ce dernier nombre est en fait obtenu par deux tirages successifs de deux entiers compris entre 1 et 6. La probabilité d'apparition n'est donc pas la même pour tous les nombres (7 par exemple « sortira » plus souvent que 2 ou que 12), et chacun aura intérêt à en tenir compte pour augmenter ses chances de gagner.

On joue en tapant n_1 R/S, puis éventuellement n_2 R/S, n_3 R/S, etc. jusqu'à ce que la somme $n_1 + n_2 + \dots$ soit égale au nombre tiré. On fait alors SBR 0. Le ou les chiffres n_1, n_2, \dots sont remplacés par des 0 dans la suite 1234567 qui constitue la partie fractionnaire de la mantisse et l'affichage indique le nombre des essais restants, les chiffres disponibles et le nouveau tirage, par exemple 8.1034067 10 si l'on a utilisé 2 et 5 au premier essai.


Exemple de partie :

10 STO 0
0.1234567 STO 6
3.01837 STO 4
3.7183037 STO 3
0.45623 STO 1

Touches pressées Affichage

SBR 1	9.1234567 08
1 R/S	9.1234567 07
3 R/S	9.1234567 04
4 R/S	9.1234567 00
SBR 0	8.0200567 11
5 R/S	8.0200567 06
6 R/S	8.0200567 00
SBR 0	7.0200007 07

etc.



Au cours de la partie, il peut arriver qu'il n'y ait aucune combinaison de chiffres permettant d'obtenir le nombre tiré, on tape alors SBR 1 pour continuer.

On devra veiller à ne jamais entrer deux fois le même chiffre et à ne pas tricher en utilisant un chiffre déjà « remis à zéro ». Le contrôle effectué par le programme porte seulement sur le fait que la somme des chiffres enlevés correspond bien au nombre tiré au début.

Si le joueur épuise ses neuf essais sans parvenir à remettre à zéro la suite « 1234567 », un message clignotant est lancé (à lire à l'envers). Même chose si le joueur gagne.

Le programme, qui utilise plusieurs des techniques précédemment décrites, a été traité sous forme modulaire. Il comporte en effet quatre parties indépendantes qui pourraient être organisées différemment. Ce type de conception, difficile à pratiquer sur TI-57 (faute

de place), pourra être repris avantageusement sur des appareils plus importants. On aura ainsi un module d'affichage, un module de contrôle, un module pour l'essai du joueur, etc.

Comme vous le verrez en pratiquant ce jeu, on ne gagne pas à tous les coups, mais il ne s'agit pas uniquement d'un jeu de hasard : attachez-vous à rechercher quelle est la meilleure stratégie à adopter.

□ Jacques Deconchat